
1 @ARUN:'RED ALERT Init 

10 VIDEO3:CLP:OUT1,-2 

20 POKE#700,54,65,65,0,65,65,54,0,0,65,65,54,64,64,0,0,48,73,73,0,

73,73,6,0,0,73,73,0,73,73,54,0,6,9,9,9,54,8,8,0 

30 POKE#728,6,73,73,0,73,73,48,0,54,73,73,0,73,73,48,0,0,1,1,48,9,

9,6,0,54,73,73,0,73,73,54,0,6,73,73,0,73,73,54,0 

40 POKE#780,54,65,65,0,73,73,48,0,54,9,9,0,9,9,54,0,54,1,1,6,1,1,5

4,0,54,73,73,0,73,73,0,0 

50 POKE#7A0,54,65,65,0,65,65,54,0,54,64,64,48,8,8,6,0,54,73,73,0,7

3,73,0,0,118,9,9,48,73,73,6,0 

60 POKE#7C0,48,8,8,6,1,1,0,0,8,42,0,99,0,42,8,0,3,6,143,255,143,6,

3,0,0,0,60,126,60,0,0,0 

70 GOTO520 
 

100 CLS:CLV 

110 Y=3:LET[0],0,3,0,1,16,3,0,12,0,0,16,3 

120 FORA=8TO10:LC12,A:?CHR$(224);:NEXT 

130 LCX,Y:?CHR$(251);:LC4,8:?CHR$(224);:LC4,9:?CHR$(248);:LC4,10:?

CHR$(227); 

140 PLAY"T500DRG<C>BA<D2":WAIT70 

150 LRUN1 
 

500 PLAY"O5T500C+C1": 

510 FORZ=0TO1:FORA=0TO14STEP2:FORB=1TO5STEP2:LCA,B:?CHR$(249*!Z);:

NEXT:NEXT:NEXT 

520 FORA=0TO1:FORB=0TO3:LC10-A*2,B+2:?CHR$(240+A*4+B);:NEXT:NEXT:C

LK 

530 IF!IN(4)GOTO100ELSECONT 

 
※ RED ALERT InitはSAVE 0でスロットの 0番にセーブして下さい。 
 

 

1 'RED ALERT Main 

100 L=!IN(1):R=!IN(2):IF!IN(5)*PF=1:BEEP9*!X 

110 LCX,Y:?CHR$(0);:IFLorRY=Y-(L*(Y>1)-R*(Y<5))*2ELSEY=3 

120 X=X+F*2:IFX=16F=0:I=(C+10)%60:P=0:X=15 

130 IFI=CP=1:X=0 

140 LCX,Y:IF(X=[4])*(Y=[5])W=0:GSB410ELSEIF(X=[10])*(Y=[11])W=6:GS

B410ELSE?CHR$(251*P); 
 

200 IF[3+W]*C%2GOTO260 

210 LC[4+W],[5+W]:?CHR$(0);:IF![4+W]BEEP80,6:WAIT5:D=D+1:GSB400:IF

[5+W]=3D=3 

220 LC4,8:?CHR$(224+D);:IFD=3LRUN0,500 

230 IF![W]*[2+W][W]=1:[5+W]=1+RND(3)*2 

240 A=[5+W]:IF!RND(5)B=RND(3)-1:[5+W]=A+((A<5)-(A>1)+(A=3)*B)*2 

250 [4+W]=[4+W]-[W]*2 

260 A=[4+W]:LCA,[5+W]:IF!(SCR(A,[5+W])=251)?CHR$(250*[2+W]*!(A=1

6));:GOTO280 

270 GSB410 

280 LC12,8:?CHR$(224+S/100);:LC12,9:?CHR$(224+S%100/10);:LC12,10:?

CHR$(224+S%10);: 

290 IF[1+W]=C[2+W]=1:[3+W]=RND(2) 

300 W=(W+6)%12:IFWGOTO200 

310 C=(C+1)%60:BEEP90*([2]|[8]):WAIT5:GOTO100 
 

400 [W]=0:[1+W]=(C+RND(12)+12)%60:[2+W]=0:[4+W]=16:RTN 

410 BEEP19,9:?CHR$(249);:WAIT5:I=(C+1+!F*9)%60:F=0:P=0:S=S+X/2:X=1

5:GSB400:?CHR$(28,0);:RTN 

 

※ RED ALERT MainはSAVE 1でスロットの 1 番にセーブして下さい。 
※ リセット後自動起動します。停止させた場合は必ずRED ALERT Initから実行して下さい。 
 
 
 

By nana_hachimaru@X(Twitter) 



■ 処理の内訳 ■ 
 
RED ALERT Init 
10～70 行 ： 画面設定、PCG 定義（20～30 行：数字、40～50 行：「GAMEOVER」、60 行：「/」、爆発、クリンゴン艦、光子魚雷） 
100～150 行 ： 変数定義、フィールド描画、OP 曲 
500～530 行 ： ED 曲、爆発～GAMEOVER、リプレイ待ち 
 
RED ALERT Main 
100～140 行 : 自弾処理 
200～310 行 : 敵機処理 
400 行 : 敵機変数リセット処理 
410 行 : 撃破処理 
 
■ 配線 ■ 
 

パーツ IchigoJam 

左移動ボタン IN1(CN4) GND(CN4) 

右移動ボタン IN2(CN4) GND(CN4) 

スタートボタン IN4(CN4) GND(CN4) 

ショットボタン OUT1(CN4) GND(CN4) 

スピーカー SOUND(CN3) GND(CN3) 

昇圧モジュール入力側 5V(CN5)  GND(CN5) 

モニター 映像出力、昇圧モジュールの出力 
 

 

 

 

 

 

 

 

 

 
 
 

 

 
 

 
■ 使用パーツ ■ 
 
IchigoJam S 
ケース ： TAKACHI YM-180(W180×H40×D130) 
昇圧モジュール ： 電子工作ステーション DCDC 昇圧コンバーターモジュール（5V,8V,9V,12V） 
 
その他 
4.3 インチバックビューモニター（ジャンク）、スピーカー（ジャンク）、モーメンタリスイッチ 4個、RCA コネクター（オス-オス）、 
スペーサー（M3*6）、ジャンパーワイヤー（片側オス）8 本、リード線、色透明下敷き（赤、スモーク）、テント補修テープ 
 
■ おわりに ■ 
 
「輝点のあるジャンクのモニター、何かに使えないか？」と思ったところから始まった今回の製作。 
往年の LED 電子ゲームを IchigoJam の拡大表示機能を使えば模倣できるのでは？と試みに作ったのが『あの電』です。 
ソフトの目処が立ち次は筐体。概ね手元にあるパーツで作ろうという思いから、新規に購入したのは本体のケースと昇圧モ
ジュールのみです。IchigoJam 本体は当初は旧い U型で進めていましたが、昇圧が安定せず途中で S型に替えました。 
ケースのレイアウトは各パーツを実際に配置しマーキング、ドリルと手ヤスリで加工。なので図面などがありません。 
完成後、思いつきで筐体にドット絵のマーキーなどを入れて（行き当たりばったりでの）完成となりました。 
『RED ALERT』というタイトルは元にしたゲームが『Battlestar Galactica Space Alert』（宇宙空母ギャラクチカ）だったので、
リスペクトを込めスタートレックで行こうと、ドット絵やサウンドを起こしファンにはお馴染みの「RED ALERT」としました。 
 
最後に今年 2026 年 8 月末に閉店となったパーツショップ「タカオ電子」様、お世話になりました。ありがとうございました。 
 

By nana_hachimaru@X(Twitter) 


